# delimit ;
clear ;
set more off ;
est drop _all ;
cd "R:/personlig/fenellac/menarche_replication/analysis/" ;

global plusdir "R:/personlig/fenellac/stata_ado/plus/" ;
sysdir set PLUS $plusdir ;
adopath ++ $plusdir ;

* **************************************************************************** ;
* This code replicates Appendix G, Table 3.
*
* Paper: "Age of Marriage and Women's Political Engagement: Evidence from India"
* Authors: Fenella Carpena, Francesca Jensenius
* 
* Code by Fenella Carpena
* Last update: March 3, 2020
* **************************************************************************** ;

****************************************************************************** ;
* select the sample
****************************************************************************** ;

use "./input/eligible-women-ihds02-lhs.dta", clear ;
merge 1:1 IDHH PERSONID using "./input/eligible-women-ihds02-rhs.dta" ;
tab _merge ;
assert _merge == 3 ;
drop _merge ;

* keep only rural women ;
keep if rural == 1 ;

* keep only women who are married only once (i.e., have not ever remarried) ;
drop if ever_remarried == 1 ;
  
* keep only women age of menarche between 11-18 (1st-99th pctile) ; 
summ menarche_age ;
keep if menarche_age >= 11 & menarche_age <= 18 ;

* generating district identifier ;
egen dt_id = group(STATEID DISTID) ;

****************************************************************************** ;
* label variables for the output table 
****************************************************************************** ;

label var majorsay_cook "\shortstack[l]{What to \\ cook on a \\ daily basis}" ;
label var majorsay_purchase "\shortstack[l]{Buying \\ expensive \\ items}" ;
label var majorsay_children "\shortstack[l]{How many\\ children \\ you have}" ;
label var majorsay_sick "\shortstack[l]{What to \\ do if you \\ fall sick}";
label var majorsay_buy_land "\shortstack[l]{Buying  \\ land or \\ property}";
label var majorsay_social "\shortstack[l]{Spending \\ on social \\ functions}";

local hh_decisions 
"majorsay_cook majorsay_purchase majorsay_children
 majorsay_sick majorsay_buy_land majorsay_social" ;

label var marriage_age "Marriage Age" ;

****************************************************************************** ;
* TABLE: IV regressions on mobility
****************************************************************************** ;
est drop _all ;

foreach var of varlist `hh_decisions' { ;
 	ivreghdfe `var' height_measure hindu muslim sc st obc age mother_yrs_educ father_yrs_educ (marriage_age = menarche_age), a(dt_id) robust ;
	est sto `var' ;
	estadd local distFE "Yes" ;
} ;

esttab * using "./output/table-IV-hh-decisions.tex",
	replace msign(--)
	drop(height_measure hindu muslim sc st obc age mother_yrs_educ father_yrs_educ) 
	cells(b(label() star fmt(%9.3f %9.3f)) se(par)) 
	star(* 0.10 ** 0.05 *** 0.01) 
	stats(N, fmt(%9.0f) labels("Observations"))
	prehead(\begin{table}[p]\centering \normalsize `"\def\sym#1{\ifmmode^{#1}\else\(^{#1}\)\fi}"'
		\captionsetup{justification=centering} 
		\caption{Effects on Household Decision-Making, Rural Women}
		\label{table-IV-hh-decisions}
		\begin{tabular*}{\hsize}{@{\hskip\tabcolsep\extracolsep\fill}l*{@E}{p{1.5cm}}}
		\toprule)
	legend label  booktabs  collabels( , none)
	mgroups("Respondent has major say in", pattern(1 0 0 0 0 0 0 0) prefix(\multicolumn{@span}{c}{) suffix(}) span erepeat(\cmidrule(lr){@span}))
	postfoot(`"\bottomrule"'  \end{tabular*} \captionsetup{justification=justified, width=\hsize} 
		\caption*{\small \textit{Notes:} 
		IV regression, \textit{Marriage Age}instrumented with \textit{Menarche Age}.
		Data from IHDS 2011-12, rural women.
		All regressions include district FEs, height, age, parents' education, caste, religion. 
		Robust SEs in parenthesis. ***\$\,p < 0.01$, **\$\,p < 0.05$, *\$\,p<0.10$.\looseness=-1}
		\end{table}) ;

exit ;


